

УДК 629.113

САХНО В.П., д.т.н., профессор
Национальный транспортный университет;
ВЕРБИЦКИЙ В.Г. д.ф.-м.н., профессор
Государственный экономико-технологический университет транспорта;
КУПЛИНОВ А.В., аспирант
Национальный транспортный университет;
ЛЫСЕНКО А.А., аспирант
Национальный транспортный университет

ЧИСЛЕННЫЙ АНАЛИЗ ПРОСТРАНСТВЕННОЙ МОДЕЛИ 4-Х КОЛЕСНОГО ЭКИПАЖА В СРЕДЕ ПАКЕТА MAPLE

Приводится листинг программы численного анализа динамических качеств пространственной маятниковой модели 4-х колесного экипажа, которая включает учет взаимовлияния поперечных и продольных сил, влияние перераспределения вертикальных реакций на параметры стационарных режимов движения и их устойчивость.

Ключевые слова: колесный экипаж, численное моделирование, устойчивость движения

Введение

Один из наиболее распространенных подходов проверки непротиворечивости результатов, полученных на основе аналитических методов исследования [1,2], состоит в использовании численных методов. Системы численно-аналитических преобразований [3,4] предоставляют в этом случае весь необходимый спектр средств исследования – итерационные методы поиска стационарных режимов, получение соответствующей линеаризованной системы, нахождение корней характеристического уравнения, построение интегральных кривых и фазовых портретов системы. Полученные на этом пути результаты помогают определить границы «работоспособности» модели (ее адекватности), установить преимущество между последовательностью уточняющихся моделей.

Основная часть

Модель представлена в виде двух сочлененных частей (несущей неподдрессоренной и кренящейся подрессоренной), соединенных цилиндрическим шарниром (ось шарнира горизонтальна – совпадает с продольной осью симметрии неподдрессоренной платформы). Точка крепления цилиндрического шарнира совпадает с центром инерции несущей платформы C_o ; в невозмущенном положении центр инерции подрессоренной части C проектируется в точку C_o . Расстояние между точками C_o и C – h . При крене корпуса, возникает восстанавливающий момент пропорциональный углу крена φ (phi) $K\varphi$ – коэффициент пропорциональности (суммарная жесткость). Массы и моменты инерции частей: $m_1, m_2, J_{1zz}, J_{2xx}, J_{2xy}, J_{2yy}, J_{2xz}, J_{2yz}, J_{2zz}$. Скорость изменения курсового угла P_s (ψ – курсовой угол), скорость изменения угла крена φ (Phi) влияет на полную вертикальную реакцию от опорной поверхности; u – поперечная составляющая скорости точки C_o , а v – ее продольная составляющая; Y_i – сведенные боковые силы на передней и задней осях экипажа, a, b – расстояние точки C_o до передней и задней осей соответственно. Уравнение e1 отвечает переменной u ; e2 – ω ; e3 – φ .

restart:

delta11:=theta-(u+a*omega)/(v-H*omega);

delta12:=theta-(u+a*omega)/(v+H*omega);

```

delta21:=(-u+b*omega)/(v-H*omega);
delta22:=(-u+b*omega)/(v+H*omega);
C10:=19;C20:=19;
Y11:=C11*delta11/sqrt(1+(C10*delta11/(fi11))^2);
Y12:=C12*delta12/sqrt(1+(C10*delta12/(fi12))^2);
Y21:=C21*delta21/sqrt(1+(C20*delta21/(fi21))^2);
Y22:=C22*delta22/sqrt(1+(C20*delta22/(fi22))^2);
C11:=-0.0012*Z11^2+19*Z11;
C12:=-0.0012*Z12^2+19*Z12;
C21:=-0.0012*Z21^2+19*Z21;
C22:=-0.0012*Z22^2+19*Z22;
Z11:=Nz*b/(2*I)-Ny*h1*b/(I*(2*H))-(Kphi*phi+BPhi*Phi)*b/(I*(2*H));
Z12:=Nz*b/(2*I)+Ny*h1*b/(I*(2*H))+(Kphi*phi+BPhi*Phi)*b/(I*(2*H));
Z21:=Nz*a/(2*I)-Ny*h1*a/(I*(2*H))-(Kphi*phi+BPhi*Phi)*a/(I*(2*H));
Z22:=Nz*a/(2*I)+Ny*h1*a/(I*(2*H))+(Kphi*phi+BPhi*Phi)*a/(I*(2*H));
fimax:=0.87;
fi11:=fimax;
fi12:=fimax;
fi21:=fimax*sqrt(1-(X21/Z21)^2);
fi22:=fimax*sqrt(1-(X22/Z22)^2);(26)
X21:=Z21/(Z21+Z22)*((m1+m2)*g*(0.0041+0.0041*v/44.7+u*omega/g)+0.5*ro*v^2*Cx*A);
X22:=Z22/(Z21+Z22)*((m1+m2)*g*(0.0041+0.0041*v/44.7+u*omega/g)+0.5*ro*v^2*Cx*A);
Nz:=(m1+m2)*g-m2*Phi^2*h*cos(phi);
Ny:=omega*(m1+m2)*v;
e1:=subs({sin(phi)=0,cos(phi)=1},m1*U+m2*U-m2*PP*h*cos(phi)+m2*Phi*h*sin(phi)*Phi+
+omega*m1*v+omega*m2*v+m2*omega^2*h*sin(phi)-(Y11+Y12)*cos(theta)-(Y21+Y22));
e2:=subs({sin(phi)=0,cos(phi)=1},J1zz*Omega+J2yy*Omega*sin(phi)^2+J2zz*Omega*cos(phi)^2-
-J2xy*PP*sin(phi)-J2xz*PP*cos(phi)-2*J2yz*Omega*sin(phi)*cos(phi)+m2*Omega*h^2*sin(phi)^2-
-m2*h*(u*omega*sin(phi)+v*Phi*cos(phi))-a*(Y11+Y12)*cos(theta)+b*(Y21+Y22));
e3:=subs({sin(phi)=0,cos(phi)=1},J2xx*PP-J2xy*Omega*sin(phi)-J2xz*Omega*cos(phi)-m2*h*cos(phi)*U+
+m2*h^2*cos(phi)^2*PP-2*m2*h^2*cos(phi)*Phi*sin(phi)*Phi-J2yy*omega^2*sin(phi)*cos(phi)+
+J2zz*omega^2*cos(phi)*sin(phi)+2*J2yz*omega^2*cos(phi)^2-J2yz*omega^2+m2*Phi^2*h^2*sin(phi)*cos(phi)-
-m2*omega*h*cos(phi)*v-m2*omega^2*h^2*cos(phi)*sin(phi)+Kphi*phi+BPhi*Phi-m2*g*h*phi);
with(linalg,jacobian);
with(linalg);
ss:=solve({e1,e2,e3},{U,Omega,PP});
assign(ss);
U1:=subs({u=u(t),omega=omega(t),phi=phi(t),Phi=Phi(t)},U);
Omega1:=subs({u=u(t),omega=omega(t),phi=phi(t),Phi=Phi(t)},Omega);
P1:=subs({u=u(t),omega=omega(t),phi=phi(t),Phi=Phi(t)},PP);
Z22:=subs({u=u(t),omega=omega(t),phi=phi(t),Phi=Phi(t)},Z22);
Z21:=subs({u=u(t),omega=omega(t),phi=phi(t),Phi=Phi(t)},Z21);
fi21:=subs({u=u(t),omega=omega(t),phi=phi(t),Phi=Phi(t)},fi21);

```

Числовые значения параметров:

```

v:=10;theta:=0.2;
g:=9.81;H:=0.8;
h:=0.45;h1:=0.3;
Kr:=1/30;ro:=1.3;A:=2.2;Cx:=0.5;
m1:=200;m2:=830;
l:=2.59;a:=1.06;b:=l-a;

```

J1zz:=0;J2xx:=500;J2yy:=0;J2zz:=1600;J2xz:=30;J2yz:=0;J2xy:=0;
Kphi:=40000;BPhi:=2*750;

ss0:=fsolve({subs(Phi=0,U)=0,subs(Phi=0,Omega)=0,subs(Phi=0,PP)=0},{u=0,omega=0,phi=0});

Результат:

$$ss0 := \{ \phi = 0.06088993676, u = 0.1982773227, \omega = 0.5923680353 \}$$

(параметры кругового стационарного режима)

with(LinearAlgebra):

JJak:=evalf(Matrix(subs(Phi=0,jacobian([U,Omega,Phi,PP], [u,omega,phi,Phi]))));

AA := evalf(Matrix(subs(ss0, JJak)));

$$AA := \begin{bmatrix} -1.776238438 & -9.219291353 & -33.08110704 & -1.286455898 \\ 0.7620073854 & -1.865168104 & -2.301868628 & 2.245680838 \\ 0. & 0. & 0. & 1. \\ -0.9588217270 & 0.3527143409 & -72.98703671 & -2.863631859 \end{bmatrix}$$

(коэффициенты матрицы линейного приближения в окрестности найденного стационарного режима движения)

Re(Eigenvalues(AA));

$$\begin{bmatrix} -1.81641576082018 \\ -1.81641576082018 \\ -1.43610343967983 \\ -1.43610343967983 \end{bmatrix}$$

(действительные части корней характеристического уравнения)

F:=dsolve({diff(u(t),t)=U1,diff(omega(t),t)=Omega1,diff(phi(t),t)=Phi(t),diff(Phi(t),t)=P1,diff(psi(t),t)=omega(t),diff(xc(t),t)=v*cos(psi(t))-u(t))*sin(psi(t)), diff(yc(t),t)=v*sin(psi(t))+(u(t))*cos(psi(t)),u(0)=0,omega(0)=0.01,phi(0)=0,Phi(0)=0,psi(0)=0,xc(0)=0, yc(0)=0},{u(t),omega(t),phi(t),Phi(t),psi(t),xc(t),yc(t)},numeric,output=listprocedure);

```
with(plots):  
odeplot(F,[t,phi(t)],0..10);
```




Рис. 1 Интегральная кривая $\phi(t)$

```
odeplot(F,[xc(t),yc(t)],0..13);
```




Рис. 2 Траектория центра масс, отвечающая стационарному режиму движения

odeplot(F,[t,fi21],0..10);




Рис. 3 Изменение коэффициента сцепления на внутреннем колесе задней оси

odeplot(F,[[t,Z21],[t,Z22]],0..10);




Рис. 4 Изменение вертикальных нагрузок на внешнем колесе задней оси – верхний график, на внутреннем колесе – нижний график

```
odeplot(F,[[t,-0.0012*Z21^2+19*Z21],[t,-0.0012*Z22^2+19*Z22]],0..10);
```




Рис. 5 Изменение коэффициентов сопротивления уводу на внешнем колесе задней оси – верхний график, на внутреннем колесе – нижний график)

Выводы

Проведен анализ динамических качеств пространственной модели колесного экипажа на основе численных методов пакета MAPLE. Визуальное представление результатов анализа дает возможность оценить преемственность в последовательно уточняющихся моделях и определить наиболее «весомые» звенья этой последовательности.

Список литературы

1. Управляемость и устойчивость автомобиля; пер. с англ. к.т.н. В.И. Котовского; под ред. д.т.н. А.С. Литвинова. – М.: Машгиз, 1963. – 267 с.
2. Лобас Л.Г., Вербицкий В.Г. «Качественные и аналитические методы в динамике колесных машин», Киев: Наукова Думка; 1990г. – 232 с.
3. Говорухин В., Цибулин В. Компьютер в математическом исследовании. Учебный курс.— СПб.: Питер, 2001. — 624 с.
4. Погорелов Д.Ю. «Введение в моделирование динамики систем тел», Учеб пособие, Брянск, БГТУ, 1997 г, 156 с.

Сахно В.П., Вербицкий В.Г. Куплінов А.В, Лисенко О.О. Чисельний аналіз просторової моделі 4-х колісного екіпажа в середовищі пакету MAPLE

Анотація. Приводиться лістинг програми чисельного аналізу динамічних якостей просторової маятникової моделі 4-х колісного екіпажа, яка включає врахування взаємовпливу поперечних і поздовжніх сил, вплив перерозподілу вертикальних реакцій на параметри стаціонарних режимів руху і їх стійкість.

Ключові слова: колісний екіпаж, чисельне моделювання, стійкість руху

Sahno V.P., Verbitsky V.G. Kuplinov A.V, Lysenko A.A. Numerical analysis of spatial models 4-wheeled vehicle in the environment of MAPLE

***Abstract.** Provides a listing of program for numerical analysis of dynamic spatial qualities pendulum model 4 wheeled vehicle, which includes consideration of the mutual influence of the transverse and longitudinal forces, the impact of vertical redistribution reactions on the parameters of stationary regimes of motion and stability.*

Keywords: vehicle, numerical modeling, stability of motion.

Стаття надійшла до редакції 11.02.2014 р.